
 Friedrich-Ebert-Schule Esslingen	µC-Steuerungen mit Visual Basic	Name:
3.2.1	Übung 3/1: Ändern der Comport-Einstellungen	Datum:

Der Anschlußname (COMx) des Comport soll nun über eine Combo-Box eingestellt werden können!

Arbeitsauftrag




1. Ändern Sie das VB-Programm gemäss dem abgebildeten Formular und Listing und fügen Sie die Funktion `GetSerialPortNames()` ein.
2. Die Bilder für den Port-Indikator  **Open.gif** und  **Close.gif**, müssen den Projektressourcen hinzugefügt werden. (My Project → Resources)

Formular


VB-Quellcode

```
Imports System.IO.Ports 'Importiert den System.IO.Ports-Namensraum

Public Class Empfangen2
    Dim WithEvents ComPort As New SerialPort 'WithEvents erbt die Ereignisse der SerialPort-
Private Sub Form1_Load(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles
    Control.CheckForIllegalCrossThreadCalls = False ' Unterdrückt Thread-Fehler Behandlung
    GetSerialPortNames() ' Comport Settings Auswahlfeld füllen!!
    cbPortNames.SelectedIndex = 0 'Beim Start wird der erste verfügbare Comport ausgewählt
    btnOpenPort.Text = "Comport öffnen" ' Schaltflächen-Beschriftung beim Start festlegen!
End Sub
Private Sub Form1_FormClosed ...
Private Sub btnOpenPort_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) I
    If Not ComPort.IsOpen Then
        InitSerialPort()
        ComPort.Open()
        pbxPortIndikator.Image = My.Resources.Open ' Bild "Open.gif" laden
        btnOpenPort.Text = "Comport schließen"
    Else
        ComPort.Close()
        pbxPortIndikator.Image = My.Resources.Closed ' Bild "Close.gif" laden
        btnOpenPort.Text = "Comport öffnen"
    End If
End Sub
Private Sub ComPort_DataReceived ...
'***** sonstige Prozeduren *****
Private Sub InitSerialPort()
    If ComPort.IsOpen Then ComPort.Close() 'ComPort muss geschlossen sein
    ComPort.PortName = cbPortNames.SelectedItem 'Comport festlegen!
    ComPort.NewLine = vbCrLf 'Zeilenendezeichen festlegen!
End Sub
Private Sub GetSerialPortNames()
    cbPortNames.Items.Clear() 'ComboBox cbPortNames leeren
    cbPortNames.Sorted = True 'Inhalte sortiert darstellen
    'ComboBox neu einlesen
    For Each sp As String In My.Computer.Ports.SerialPortNames
        cbPortNames.Items.Add(sp)
    Next
End Sub
End Class
```

 Friedrich-Ebert-Schule Esslingen	µC-Steuerungen mit Visual Basic	Name:
3.2.2	Übung 3/2: Ändern der Comport-Einstellungen	Datum:

Nun sollen auch verschiedene andere Schnittstellen-Parameter einstellbar sein.



Arbeitsauftrag

1. Ändern Sie das Formular wie abgebildet. Fügen Sie die angegebenen Prozeduren in den Code ein.
2. Ändern Sie die Form-Load-Ereignisprozedur, so daß alle Schnittstellenparameter auf die Standardwerte (n,9600,8,n,1) eingestellt werden!!

```
cbBaudrate.Items.AddRange(New String() {"1200", "2400", "4800", "9600", "14400"})
```


Formular

VB-Quellcode

```
Private Sub InitSerialPort()
    With ComPort
        If .IsOpen Then .Close() 'ComPort muss geschlossen sein
        .PortName = cbPortNames.SelectedItem 'Der ausgewählte Comport wird geöffnet!
        .BaudRate = cbBaudrate.SelectedItem 'gewählte Baudrate wird eingestellt!
        .Parity = cbParity.SelectedItem 'gewählte Parität wird eingestellt!
        .Handshake = cbHandshake.SelectedItem 'gewählter Handshake wird eingestellt!
        .DataBits = numDatenbits.Value 'Datenbits einstellen!
        .StopBits = numStoppbits.Value 'Stoppbits einstellen!
        If rbtnCRLF.Checked Then .NewLine = vbCrLf 'für µC-printf mit \n wird zu CR+LF
        If rbtnLF.Checked Then .NewLine = vbLf 'für µC-Assembler LF = 0Ah
    End With
End Sub

Private Sub GetHandshakeModes()
    For Each hs As String In System.Enum.GetNames(GetType(IO.Ports.Handshake))
        cbHandshake.Items.Add(hs)
    Next
End Sub

Private Sub GetParityModes()
    For Each p As String In System.Enum.GetNames(GetType(IO.Ports.Parity))
        cbParity.Items.Add(p)
    Next
End Sub
```

 Friedrich-Ebert-Schule Esslingen	µC-Steuerungen mit Visual Basic	Name:
3.2.3	Übung 3: Ändern der Comport-Einstellungen (Listing)	Datum:

```
Imports System.IO.Ports 'Importiert den System.IO.Ports-Namensraum
Public Class Empfangen3
    Dim WithEvents ComPort As New SerialPort 'WithEvents erbt die Ereignisse der SerialPort-
                                           'Klasse
    '***** Ereignisprozeduren *****
    Private Sub Form1_Load(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles MyBase.Load
        Control.CheckForIllegalCrossThreadCalls = False ' Unterdrückt Thread-Fehler Behandlung
        ' Comport Settings Auswahlfelder füllen!!
        GetSerialPortNames()
        cbBaudrate.Items.AddRange(New String() {"1200", "2400", "4800", "9600", "14400"})
        GetParityModes()
        GetHandshakeModes()

        'Comport Settings Startwerte vorbereiten!
        cbPortNames.SelectedIndex = 0 'Beim Start wird der erste verfügbare Comport ausgewählt
        cbBaudrate.SelectedIndex = 3 'Baudrate = 9600
        numDatenbits.Value = 8 '8 Datenbits
        cbParity.SelectedIndex = 0 'Parität = keine
        numStoppbits.Value = 1 '1 Stoppbit
        cbHandshake.SelectedIndex = 0 'Flusssteuerung = keine

        btnOpenPort.Text = "Comport öffnen"
    End Sub

    Private Sub Form1_FormClosed(ByVal sender As Object, ByVal e As System.Windows.Forms.FormClosedEventArgs) Handles Me.FormClosed
        If ComPort.IsOpen Then ComPort.Close() 'ComPort wird sicher geschlossen
    End Sub

    Private Sub btnOpenPort_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles btnOpenPort.Click
        If Not ComPort.IsOpen Then
            InitSerialPort()
            ComPort.Open()
            pbxPortIndikator.Image = My.Resources.Open
            btnOpenPort.Text = "Comport schließen"
        Else
            ComPort.Close()
            pbxPortIndikator.Image = My.Resources.Closed
            btnOpenPort.Text = "Comport öffnen"
        End If
    End Sub

    Private Sub ComPort_DataReceived(ByVal sender As Object, ByVal e As SerialDataReceivedEventArgs) Handles ComPort.DataReceived
        tbReceive.Text = ComPort.ReadLine()
    End Sub

    '***** sonstige Prozeduren *****
    Private Sub InitSerialPort()
        With ComPort
            If .IsOpen Then .Close() 'ComPort muss geschlossen sein
            .PortName = cbPortNames.SelectedItem 'Der ausgewählte Comport wird geöffnet!
            .BaudRate = cbBaudrate.SelectedItem 'gewählte Baudrate wird eingestellt!
            .Parity = cbParity.SelectedItem 'gewählte Parität wird eingestellt!
            .Handshake = cbHandshake.SelectedItem 'gewählter Handshake wird eingestellt!
            .DataBits = numDatenbits.Value 'Datenbits einstellen!
            .StopBits = numStoppbits.Value 'Stoppbits einstellen!
            If rbtnCRLF.Checked Then .NewLine = vbCrLf 'für µC-printf mit \n wird zu CR+LF
            If rbtnLF.Checked Then .NewLine = vbLf 'für µC-Assembler LF = 0Ah
        End With
    End Sub

    Private Sub GetSerialPortNames()
        cbPortNames.Items.Clear() 'ComboBox cbPortNames leeren
        cbPortNames.Sorted = True 'Inhalte sortiert darstellen
        'ComboBox neu einlesen
        For Each sp As String In My.Computer.Ports.SerialPortNames
            cbPortNames.Items.Add(sp)
        Next
    End Sub

    Private Sub GetHandshakeModes()
        For Each hs As String In System.Enum.GetNames(GetType(IO.Ports.Handshake))
            cbHandshake.Items.Add(hs)
        Next
    End Sub

    Private Sub GetParityModes()
        For Each p As String In System.Enum.GetNames(GetType(IO.Ports.Parity))
            cbParity.Items.Add(p)
        Next
    End Sub
End Class
```